

串行A/D转换器TLC1543及其应用

· 严天峰 ·

TLC1543是美国TI公司生产的多通道、低价格的模数转换器。采用串行通信接口，具有输入通道多、性价比高、易于和单片机接口的特点，可广泛应用于各种数据采集系统。TLC1543为20脚DIP封装的CMOS 10位开关电容逐次A/D逼近模数转换器，引脚排列如图1所示。其中A0~A10（1~9、11、12脚）为11个模拟输入端，REF+（14脚，通常为VCC）和REF-（13脚，通常为地）为基准电压正负端，CS（15脚）为片选端，在CS端的一个下降沿变化将复位内部计数器并控制和使能ADDRESS、I/O CLOCK（18脚）和DATA OUT（16脚）。ADDRESS（17脚）为串行数据输入端，是一个4位的串行地址用来选择下一个即将被转换的模拟输入或测试电压。DATA OUT 为A/D转换结束3态串行输出端，它与微处理器或外围的串行口通信，可对数据长度和格式灵活编程。I/O CLOCK为数据输入/输出提供同步时钟，系统时钟由片内产生。芯片内部有一个14通道多路选择器，可选择11个模拟输入通道或3个内部自测电压中的任意一个进行测试。片内设有采样-保持电路，在转换结束时，EOC（19脚）输出端变高表明转换完成。内部转换器具有高速（10 μ S转换时间），高精度（10位分辨率，最大 \pm 1LSB不可调整误差）和低噪声的特点。

1. TLC1543的工作时序 TLC1543工作时序如图2所示，其工作过程分为两个周期：访问周期和采样周期。工作状态由CS使能或禁止，工作时CS必须置低电平。CS为高电平时，I/O CLOCK、ADDRESS被禁止，同时DATA OUT为高阻状态。当CPU使CS变低时，TLC1543开始数据转换，I/O CLOCK、ADDRESS使能，DATA OUT脱离高阻状态。随后，CPU向ADDRESS端提供4位通道地址，控制14个模拟通道选择器从11个外部模拟输入和3个内部自测电压中选通1路送到采样保持电路。同时，I/O CLOCK端输入时钟时序，CPU从DATA OUT 端接收前一次A/D转换结果。I/O CLOCK从CPU 接受10个时钟长度的时钟序列。前4个时钟用4位地址从ADDRESS端装载地址寄存器，选择所需的模拟通道，后6个时钟对模拟输入的采样提供控制时序。模拟输入的采样起始于第4个I/O

CLOCK的下降沿，而采样一直持续6个I/O CLOCK周期，并一直保持到第10个I/O CLOCK的下降沿。转换过程中，CS的下降沿使DATA OUT引脚脱离高阻状态并起一次I/O CLOCK的工作过程。CS的上升沿终止这个过程并在规定的延迟时间内使DATA OUT引脚返回到高阻状态，经过两个系统时钟周期后禁止I/O CLOCK和ADDRESS端。

2. 软硬件设计要点 TLC1543的三个控制输入端CS、I/O CLOCK、ADDRESS和一个数据输出端DATA OUT遵循串行外设接口SPI协议，要求微处理器具有SPI接口。但大多数单片机均未内置SPI接口（如目前国内广泛采用的MCS51和PIC系列单片机），需通过软件模拟SPI协议以便和TLC1543接口。TLC1543芯片的三个输入端和一个输出端与51系列单片机的I/O口可直接连接，具体连接方式可参见图3。

软件设计中，应注意区分TLC1543的11个模拟输入通道和3个内部测试电压地址。附表为模拟通道和内部电压测试地址。程序软件编写应注意TLC1543通道地址必须为写入字节的高四位，而CPU读入的数据是芯片上次A/D转换完成的数据。在本文后附的程序中对此有详细的说明。

附

模拟输入通道选择	输入寄存器地址（2进制）	
A0	0000	
A1	0001	
A2	0010	
A3	0011	
A4	0100	
A5	0101	
A6	0110	
A7	0111	
A8	1000	
A9	1001	
A10	1010	
<hr/>		
内部测试电压选择	输入地址	输出结果(16进制)
$(V_{ref+} - V_{ref-})/2$	1011	200
V_{ref-}	1100	000
V_{ref+}	1101	3ff
注： V_{ref+} 为加到TLC1543 REF+端的电压， V_{ref-} 是加到REF-端的电压		

3. TLC1543的应用 图3为TLC1543在一个通信电源数据采集系统中的实际应用。此例中的TLC1543主要用于完成8组-48V直流电源、2组220V交流电源和1组温度参数的采样。采样数据由89C51单片机通过RS232标准串口送给后台PC机进行处理。各单元的功能介绍如下：

看门狗和E²PROM存储器 图3中的X25045（IC4）是一种新型的看门狗和E²PROM存储芯片，它将电压监控、看门狗定时器和E²PROM三种功能组合在单个芯片之内。X25045还为89C51提供上电复位，当程序紊乱或电压失常时启动内部的看门狗电路以强制单片机复位，使程序从头开始执行。X25045还内置512字节E²PROM存储单元，可随时保存各种重要数据，如A/D采样结果等，这使系统掉电后重要数据仍然不会丢失。

串行显示驱动器 PS7219 PS7219（IC2）是多功能8位LED显示驱动芯片。接口采用三线SPI方式，用户只需简单修改内部相关的控制字，便可以实现多位LED显示。图3中的PS7219主要用来显示设置参数，如显示通信速率参数等。

PC机通信接口电路 MAX232（IC3）为标准RS232接口转换芯片，主要完成TTL至RS232电平的转换，为单片机和PC机通信提供通路。在整个数据采集系统中，PC机除了处理各种采样数据外，还负责对前台单片机系统进行管理，如故障诊断，参数设置等等。参数设置的其中一项为系统通信速率设置，管理人员可通过PC机任意设置单片机和PC机的通信速率，其设置参数保存在X25045的E²PROM存储单元中，在下次设置之前，该参数不会被更改。本例为单个采集系统的应用实例，实际应用中往往存在多系统并存的情况，这时可将MAX232更换为MAX485接口芯片，采用485总线标准，通过一台PC机可在几千米范围内管理数十台前端机。

极性转换电路 鉴于目前国内采用的通信电源均为负电压，而TLC1543模拟通道输入只能为正电压，因此-48直流电压在送到A/D转换器前除了要分压外，还需将负电压转换为正电压。图4为一种简单的极性转换电路，仅增加两个电阻便可完成负电压到正电压的转换，省去了复杂的极性转换芯片。图4中当输入电压为0V时，TLC1543 A0端电压为2.5V；当输入电压为

-5V 时，A0 端电压为 0V。通过电阻 R1、R2 的简单分压便完成了输入通道的负电压到正电压的极性转换。

TLC1543 与 89C51 接口程序 TLC1543 与 89C51 接口程序应完全依照 TLC1543 的工作时序编写，主要由 CONVETER 子程序组成。由于转换完成的数据为 10 位，软件编写时将数据的高位字节存放在 2EH 单元中，低位字节存放在 2FH 单元中。其中 R4、R3 寄存器分别存放 TLC1543 的通道地址和数量；R1、R2 寄存器存放 A/D 转换结果。本例程序中 89C51 晶振为 11.0592MHz。

其程序清单如下：

```

    CLK EQU P1.4
    ADDR EQU P1.5
    DATA1 EQU P1.6
    CS EQU P1.7
    ORG 0000H
    AJMP MAIN
    ORG 0100H
MAIN:MOV SP,#60H
    MOV A,#0FFH
    MOV R4,#00h ; A/D 通道地址初值，即A0通道
    MOV R5,#15 ; 采样的通道总数，因A/D转换为上次数据，故设置15个
    MOV R0,#30H ; 采样数据存放地址
    MOV R1,#31H ;
REPEAT:MOV P1,#04h
    CLR CLK ; I/O CLOCK置低
    SETB CS ; CS 置高
    SETB DATA1
    MOV A,#0FFH
    LCALL CONVTER ; 调用转换子程序
    MOV @R0,2fh ; 转换后数据存放在以R0、R1间接寻址的寄存器中
    MOV @R1,2eh
    MOV A,R4 ; 通道地址加1，请注意是高4位加1
    ADD A,#10H
    XCH A,R4
    INC R0 ; 数据寄存器同时加2
    INC R0
    INC R1
    INC R1
    LCALL DELAY ; 调延时子程序
    DJNZ R5,REPEAT ; 所有通道转换结束后，循环等待，否则转REPEAT继续执行
    AJMP $
; TLC1543转换子程序，MSB在先
CONVTER:PUSH ACC
    CLR CS
    MOV A,R4
    NOP
    MOV R3,#8
LOOP: MOV C,data1
    RLC A
    MOV ADDR,C
    SETB CLK ; 一个I/O CLOCK时钟
    NOP

```

```

CLR CLK
DJNZ R3, LOOP
MOV C, DATA1
MOV B. 1, C ; 数据第9位存放到B. 1
SETB CLK
NOP
CLR CLK
MOV C, DATA1
MOV B. 0, C ; 数据第10位存放到B. 0
SETB CLK
NOP
CLR CLK
CLR DATA1
; 采样数据低位存放到2fh单元
RL A
RL A
MOV C, ACC. 1
MOV B. 2, C
MOV C, ACC. 0
MOV B. 3, C
MOV C, B. 1
MOV ACC. 1, C
MOV C, B. 0
MOV ACC. 0, C
MOV 2fh, A ;
; 采样数据高位存放到 2eh单元
MOV C, B. 2
MOV B. 1, C
MOV C, B. 3
MOV B. 0, C
ANL B, #00000011B
MOV 2eh, B
POP ACC
RET
; 延时子程序
DELAY:MOV 51h, #200
DELAY1:DJNZ 51h, DELAY1
RET
END

```

由TLC1543构成的上述通信电源数据采集系统做为一个大型数据监控系统
的分系统已成功地应用在某通信领域，实践证明，该系统具有采样精度高、
转换速度快的特点，具有较高的性价比，可广泛适用于各种数据采集领域。

小知识 SPI (Serial Peripheral Interface) 总线是由美国 Motorola 公司发明的一种串行总线技术，采用三线串行通讯接口，可实现芯片间的串行数据传输。采用 SPI 总线的单片机我们并不少见，如 8031 就是通过接收 (RXD)、发送 (TXD) 以及公共端 (GND) 实现数据的串行传送。

随着单片机在各种电子产品中的广泛应用，各种专用的单片机不断问世，由于制造成本等原因，它们的外围接口电路都比较简单，比如许多四位单片机都没有并行外部总线接口，普遍都只有 SPI 总线接口。因此这类单片机在外围器件的选择以及与其它单片机构成多机系统等方面受到较大限制。I²C 总线正是为解决这一矛盾推出的。

